

Handy- Compact™

User Manual

CD-UM-820409



**Comsat General Integrated Systems, Inc.
Compact Division**

We Think Harder To Make Your Work Easier

CONTENTS

This manual describes how to operate HANDY-COMPACT. The first part serves as an overview, and assumes a working knowledge of HP-41C operation and programming. The second part demonstrates applications with examples. The Appendix serves as a reference guide.

INTRODUCTION	3
PROGRAM STRUCTURE	5
ELEMENT LIBRARY	9
INTERCONNECTIONS	14
LADDER ANALYSIS	16
MANIPULATION COMMANDS	18
CONVERSION COMMANDS	21
CHARACTERIZATIONS	23
MAPPING	26
EXAMPLES	28
1 Parallel-Coupled Filter	30
2 Microstrip Coupled Line Filter	37
3 Two-Port Analysis of Transformer Structure	42
4 Ladder Analysis of Transformer Structure	45
5 Plotting Network Response	48
6 Analysis of a Feedback Amplifier	54
7 Analysis of a Matching Network	63
8 Amplifier Design	69
APPENDIX-- Reference Instructions and Tables	84

NOTE

Neither HP nor CGIS makes any express or implied warranty with regard to this program material, including, but not limited to, the implied warranties of merchantability and fitness for a particular purpose. Neither HP nor CGIS shall be liable for incidental or consequential damages in connection with use of this program material. This material is the property of CGIS. All rights reserved. No part of this document may be used, reproduced, or displayed without permission.

WARNING

The HANDY-COMPACT ROM is easily damaged by static electricity. It should be stored in its special static protection envelope when not in the calculator port. Failure to turn off the calculator before inserting or removing any peripheral may result in damage to both the calculator and the peripheral.

INTRODUCTION

HANDY-COMPACT is an HP-41C programmable calculator-based program which provides steady-state analysis for structures which can be modeled as interconnected two-port networks. Analyses are valid from very low frequencies through the millimeter wavelengths. A standard element library is provided which contains lumped and distributed elements. Special two-port networks such as an ideal transformer and voltage- and current-controlled sources are included for special applications. In addition to the element library, the program can accept measured one- or two-port data, allowing analysis and characterization of structures containing active devices. All internal calculations are based on scattering parameters, although Z, Y, G, and H parameter conversions are provided in case you prefer them.

Background

The routines contained in HANDY-COMPACT are extensions of those described in Microwave Circuit Design Using Programmable Calculators by Medley and Allen, published by Artech House. Although similar in the way they are run, the published routines differ from those in HANDY-COMPACT. HANDY-COMPACT provides increased accuracy (especially at the lower frequencies), ease of use, and execution speed. Although the book mentioned above is not required for using HANDY-COMPACT, it provides background information and an overview of CAD techniques in general.

Hardware

HANDY-COMPACT is provided in the form of a plug-in ROM for the HP-41C or HP-41CV calculator. Unlike previous calculator-based programs, the entire program is resident in the ROM and does not impact program or data memory. Additionally, networks can be analyzed without additional programs. A complex network may be modeled and analyzed across a range of frequencies without having to make key entries during the run.

HANDY-COMPACT can run on the basic HP-41C calculator. Before running HANDY-COMPACT, the calculator memory must be sized to 035 (see page 73 of the HP-41C manual). The remaining memory (28 registers) of the basic HP-41C calculator may then be used for either data storage or program memory. When using the HP-41C calculator with single memory modules, additional memory will allow you to store the element values and measured data and a circuit file for automatic analysis of complex structures. No more than two memory modules may be used when using the HP-82143A printer. However, there is no limit to the size of the circuit structure when using just the basic calculator— but you will be required to key-in the element values and measured data for each two-port when memory is not sufficient to store this information.

For added flexibility, the HP-41CV or HP-41C with the Quad-Memory Module (HP-82170A) is recommended. The HP-82143A printer is recommended for use with either system.

PROGRAM STRUCTURE

HANDY-COMPACT provides you with over 85 functions, including the element library, interconnections, mappings, and characterizations. Since there are so many functions and almost as many applications, no key assignments are made. You may design your own custom keyboard to fit specific applications. For example, for microwave applications, you might include the standard interconnections, all distributed elements, and a few of the lumped elements which usually occur in the form of parasitics. For low frequencies, you will not require the distributed elements. It is suggested that the card reader function "WSTS" (write status) be used to store the key assignments on a magnetic card. In fact, several keyboards can be designed for different applications.

Analysis Modes

Two analysis modes are featured, two-port and ladder. The two-port analysis mode is intended for the analysis of networks which consist of interconnections of two-ports, while the ladder analysis mode provides a much faster means of analysis of simple one-port or ladder structures. In the two-port analysis mode, two two-ports may be in use at any time. These two-ports are called two-port "A" and two-port "B", and each has a specific purpose. All library elements are placed in "B" while the interconnections connect "A" and "B", the result is then placed in "A". Thus, another element may be calculated, automatically stored in two-port B, then connected with two-port A. Two-port "A" is contained in data registers 01 through 08, and two-port "B" in registers 11 through 18. The S-parameters are stored as magnitude of S11, S12, S21 and S22 in registers 01 through 04, respectively. The angles of S11, S12, S21, and S22 are in 05 through 08, respectively. The data for two-port "B" is offset by 10 from that of "A", and is in exactly the same order. (These and other memory locations are summarized in the Appendix Tables.)

Using the Appendix Tables

NOTE

You should read this text for a discussion of the available options, then use the Appendix tables for actual program operation and register reference.

The conventions used in this manual are that all data registers are shown circled, and the input and output data are shown in the order entered into the calculator, or rolled down from the stack for display. The series R-L-C combination, for example, calls for you to enter R (resistance in ohms), L (inductance in nanohenrys), and C (capacitance in picofarads). Thus, you enter R, ENTER, L, ENTER, C, then XEQ "SRZ". At the time of execution of program "SRZ", the stack is: "x" = C; "y" = L; "z" = R.

The tables also show that there are five additional data registers which are set aside for the user. None of these registers are modified by the program. Data register 00 contains the reference impedance, and 20 contains the analysis frequency (GHz). Frequencies must be specified before use. The reference impedance is that impedance to which the S-parameter calculations are referenced as well as the reference used for conversions between reflection coefficient and impedance or admittance. The frequency in register 20 is the frequently used for the S-parameter calculations for frequency-dependent elements. Registers 21 and 22 are allocated for the stop and step frequencies, respectively, and register 34 is allocated for use as an index register. The

program does not use the latter registers; however, you can use them as part of the circuit file to increment the analysis frequency and recall or save two-port data. There are three flags which are frequently used for the program. These are flag 00, the ladder analysis flag; flag 01, shunt element flag; and flag 04, automatic execution. You should remember that flag 24, the calculator Range Error Ignore, is set by this program, and should be cleared for other programs which require a range error.

The following seven sections of this manual explain the corresponding seven tables in the Appendix.

Executing Functions

With each program routine, you enter the data as shown, then XEQ the function name. The result of each routine is shown.

(Although we normally call sets of program steps "programs," we sometimes call them functions, routines, subroutines, commands, or subprograms, depending upon the context.)

The method of assigning programs to calculator keys is described in the calculator's user guide.

ELEMENT LIBRARY

S-parameters for the elements shown are calculated using the frequency in register 20, then stored as two-port "B" in registers 11 through 18. These S-parameters are normalized to the impedance in register 00. The first 11 elements are lumped elements (consisting of various combinations of resistors, inductors, and capacitors).

Series and Parallel Connections

Flag 01 is used to connect the elements in either series or shunt (flag 1 set is the shunt connection). You may want to place two short programs into the calculator, "PA" (parallel) to set flag 01 for the shunt connection, and "SE" to clear flag 01 (although not necessary since each program will automatically clear flag 01 before the return). With this feature, you can simply enter the element data as shown, XEQ either SE or PA, depending on the connection, then XEQ the function name.

Another approach is to assign the series element to a key, and the shunt element to the shifted function on the same key. This technique can be implemented for each element as follows:

1. Place the calculator in program mode and enter the following program statements:

```
Ø1 SF Ø1 Ø2 XEQ "elementname"
```

2. At this point, the calculator will automatically convert statement Ø2 to

```
Ø2 XROM "elementname"
```

which indicates the program (called "elementname") is contained in the ROM. (The calculator could not find the program in its own memory, and assumes that it is in a ROM.)

3. Enter the following program statement:

```
Ø3 RTN
```

4. Leave program mode, go to program step 000, re-enter program mode, and insert the following program statement:

```
Ø1 LBL "elementname"
```

5. Assign "elementname" to the desired shifted key.

With the shifted key assigned to XEQ, the calculator will find your program before finding

the ROM program. Therefore, your program "elementname" will set flag 01, then execute the ROM program "elementname". The unshifted key (assigned to XROM "elementname") will execute (XROM) the ROM program directly. This technique may be used to add "pre-processing" to any of the programs contained in the ROM. (You may wish to change the program units or modify the elements using this technique.)

Lumped Elements

Lumped elements include R,L, and C elements and all combinations in series and parallel configurations.

Distributed Elements

Lossless distributed elements provided include a transmission line, open stub, and shorted stub. The latter two can be connected in either series or shunt. However, the shunt-connected transmission line has no meaning and cannot be used; flag 01 is ignored for this element and not cleared by the return. Data for transmission line elements are entered in terms of characteristic impedance and length (mils) in an air dielectric. The physical length in a dielectric medium may be converted to air by multiplying it by the square root of the effective dielectric constant. These distributed elements, as well as lumped elements, may be entered as negative elements. Values for negative lumped elements are entered as negative values; however, with distributed elements, only the length of the element can be

entered as negative. Although a negative impedance may be specified, an error will occur when the negative impedance is the same as the reference found in register 00. In addition to the above distributed elements, a four-port coupled line structure is also featured as part of the element library. This structure is the inhomogenous case; that is, unequal even- and odd-mode effective dielectric constants may be specified. This four-port structure is converted into a two-port network for analysis when the termination on any two-ports is specified. As the Appendix shows, the reflection coefficient at port k is stored in registers 30 and 31, while that for port l is in 32 and 33 (magnitude and phase respectively). Port k is always the lower-numbered port when the numbering convention shown by the diagram is used. The input port is always port l, and the flag which corresponds to the desired output port is set before execution. Even-mode impedance and length data is entered first, then odd-mode impedance and length. The air length is calculated as above; however, the square root of the even- and odd-mode effective dielectric constant is used to compute the even- and odd-mode length, respectively. This element is extremely useful for determining the effect of the unequal effective dielectric constants in the microstrip case. In addition, arbitrary terminations may be used to investigate the effects of fringing capacitance of an open, or the inductance associated with a short at microwave frequencies.

Additional Elements

Additional elements can be found in the library for special applications. The ideal transformer is specified using either impedance or turns ratio. This element is useful for computing the S-parameters for a two-port with a different reference impedance at each port. For example, the insertion loss of a transformer structure may be desired. The structure may be calculated referenced to the source impedance at one port, and the load impedance at the other. The insertion loss may then be determined directly from S21. Without normalization, additional calculations are required. You are also provided with current- and voltage-controlled sources, both voltage- and current-dependent. These sources are useful for certain (usually low-frequency) models, and may be specified with complex coefficients.

INTERCONNECTIONS

Five standard two-port interconnections, cascade ("CAS"), series ("SER"), parallel ("PAR"), Series Parallel ("SPC"), and parallel-series ("PSC") are offered. These interconnections connect two-port "A" with two-port "B" and place the result in two-port "A" as shown in the Appendix. In addition, a hybrid ("HYB") interconnection, commonly used in the design of microwave amplifiers, is featured. This function connects the input and output ports of "A" and "B" to the quadrature ports of two distributed couplers and places the results in two-port "A". The two identical couplers are assumed to be ideal, and have a characteristic impedance equal to the reference impedance found in register 00. The characteristics of these couplers are specified by entering the coupling in +dB at the center frequency and the center frequency.

PBR and SBR

Parallel branch ("PBR") and the series branch ("SBR") are provided for converting one-port networks into two-ports. This allows structures to be interconnected using one of the standard functions. The branches may be calculated from the input reflection coefficient of two-port "A" with a specified load termination which is entered before execution, or with flag 02 set, directly from a specified impedance. The parallel branch is particularly useful for transistor bias networks which can be modeled as a one-port.

LADDER ANALYSIS

The ladder analysis mode is active when flag 00 is set. The ladder analysis mode is three to four times faster than the two-port analysis. The same reference impedance and frequency is used, as well as the element names. Two-port "A" is not changed by any of the ladder analysis functions, and the only two-port element which will change the ladder analysis registers is the "CPL" function. You may switch between the two modes when necessary. The ladder analysis may be used with two-port analysis to reduce execution time. The impedance of any one-port network, such as a bias network, may be calculated, then the "SBR" or "PBR" function used to interconnect the branch, or the parasitics associated with an open or short for the coupled lines may be calculated in this mode. The ladder analysis mode may be used with lossless two-port networks when the input VSWR and insertion loss is desired. In this case, the input impedance may be converted to reflection coefficient, then VSWR. The insertion loss may be calculated from the familiar relationship for lossless two-port networks, $|S_{21}|^2 = 1 - |S_{11}|^2$.

Ladder analysis is limited to series or shunt elements, and any of the R-L-C elements, as well as the TRL, SST, and OST elements may be used. The data entry procedure for the ladder mode is the same as that of the two-port mode. The ladder analysis mode calculates the input impedance for a one-port network. The magnitude of the load impedance is stored in register 25, and the angle in 26. You then enter the element values, then execute the desired element. The program computes the input impedance at this

point of the network, stores the results in 25 and 26, then displays the impedance in "x" and "y". Another element may now be connected. The program does not read the values displayed in "x" and "y", and you may overwrite or convert them to reflection coefficients as desired.

MANIPULATION COMMANDS

Seven Special Commands are provided to manipulate two-ports "A" and "B".

SVA and SVB

The "SVA" and "SVB" commands (save "A" and save "B") are used to save two-ports "A" or "B". These commands are implemented by entering a number which corresponds to two-port "N", then XEQ either command. Two-port "A" or "B" will be saved in registers $27 + 8*N$ through $34 + 8*N$; note that the calculator must be sized for $35 + 8*N$ as shown in the Appendix. For example, to save two-port "A" as two-port "2", first enter 2, then XEQ "SVA". Two-port "A" will then be saved in registers 43 through 50. This option is useful in conjunction with the "USA" and "USB" commands (see below).

USA and USB

"USA" and "USB" commands (use "A" and use "B") are used in exactly the same manner as SVA and SVB. That is, you enter "N" and XEQ either "USA" or "USB" to recall the saved two-port into either "A" or "B". Since only two two-ports may be active in the calculator, these commands are used to save temporary data. For example, the input matching network may be saved while feedback is added to a transistor, then recalled and cascaded. These commands are also used in conjunction with the index register to recall measured data for frequency analysis. You first store the S-parameters by entering the data from

the keyboard, then use "SVA" or "SVB" to store the data as "N", "N+1", etc., in the order which corresponds to the frequency. During program execution, you can recall the index register (34), then "USA" or "USB" to retrieve the data. The index register is then incremented for use at the next frequency. Commonly-used data can be loaded from a data card if desired. The S-parameters may be stored for many frequencies (e.g., 1 to 10 GHz) as two-port "1" through "10". The index register can be started at any number to recall data at the desired frequency. You can even add a program to interpolate between the measured data points. You should use the above equations to determine where these data are stored, and remember not to store any other data in this location. The index register may also be used to recall one-port data, although you will have to write a small subroutine or make it a part of the circuit file. Another suggestion for the use of these commands is to insert a third memory module during program execution to save the computed S-parameters. When the analysis is complete, the data may be moved to lower location, the memory module removed and the printer reconnected to print the results.

EXC

The "EXC" command is used to exchange two-ports "A" and "B". The interconnections connect two-ports "A" and "B", and the elements from the library are stored as two-port "B". This approach works well since the network may be analyzed without unnecessary data manipulation; however, when the first element is calculated, there is no two-port "A". The "EXC" command may be used to place the first element into "A", then the second element may be calculated as "B" and connected.

EQU and CHA

The "EQU" (equate) and "CHA" (change) commands are useful for reducing execution time. The first half of a symmetrical network may be cascaded, then "EQU" used to copy "A" into "B". The "CHA" command is then used to exchange ports 1 and 2 which forms the second half of the network.

CONVERSION COMMANDS

Multi-Port Conversions

Conversions are provided to convert from two-port data obtained from a three-terminal device with one common terminal, to another common-lead configuration. The convention used for these conversions is shown in the Appendix. The input terminal is 1, the output 3, and the common lead 2 before the conversion. The common lead may be exchanged with the input ("X12"), or the output ("X23"). Although the diagram in the Appendix shows an FET which is converted from common source to common gate and common drain. In fact, a valid transformation can be made for any three terminal networks with one common terminal.

Parameter Conversions

Conversions are also provided between the S, Y, Z, G, and H parameter sets. All parameter sets are normalized and, as with the "X12" and "X23" functions, only two-port "A" is converted. The "EXC" command may be used if a conversion for "B" is desired. Four additional commands may be used to convert from reflection coefficient to impedance ("R-Z") or admittance ("R-Y"). The commands, "Z-R" and "Y-R" may be used to convert from impedance or admittance to reflection coefficient, respectively. The reference impedance in register 00 is used for these conversions with the impedance specified in ohms and the admittance in mmhos. The "Z-R" and "Y-R" commands are useful for converting the impedance output from the ladder analysis mode into reflection coefficient. You will note in the

Appendix that these commands require you to enter the magnitude, then angle for impedance, but that the output from the ladder analysis gives magnitude in "x" and angle in "y". The "x" and "y" registers must be exchanged when using the output from the ladder analysis as input to these functions.

CHARACTERIZATIONS

The functions found in this section are useful for the calculation of parameters of two-port "A". Two-ports "A" and "B" are not modified by any of the programs of this section, and may be used at any time during analysis.

CHR

The characterization command ("CHR") is used to calculate the stability factor, maximum available gain, and the simultaneous matched source and load for a network when the stability factor is greater than 1. When the stability factor is less than 1, the stability factor and maximum stable gain is returned instead. This command has an automatic execution feature to print results rather than the normal displayed output. This feature is active when flag 04 is set. When this flag is set, the program returns immediately with the calculated results in the data registers shown in the Appendix. You should test the stability factor before printing the results since the matched source and load reflection coefficients do not exist when the stability factor is less than 1; The program does not clear these registers, and the contents do not contain meaningful data.

STB

When designing with active devices which have stability factors less than 1, complete stability information is necessary. These calculations are provided with the "STB"

(stability) command, which determines the location and radius of the stability circle on either the input or output planes, and indicates by means of a code whether the stable region is inside or outside the stability circle. The feature is interactive; that is, the program will request whether the input or output plane is desired, calculate the parameters, then ask for another calculation as shown. However, the "STB" function also has an automatic execution option using flag 04 to request printed results. With this option, you can enter a code (1 for the input plane, or 2 for the output plane), and XEQ "STB". The data will then be returned in the data registers indicated in the Appendix. When using the printer, you may wish to print "INSIDE", "OUTSIDE", OR "ABSOLUTE" instead of the code returned in register 10.

GTB

The "GTP" (gain-transducer, power) function may be used to calculate the transducer gain of two-port "A" for specified load- and source-reflection coefficients. "CGC" (constant-gain circles) may be used to construct contours of constant gain on the input or output plane. "MMC" (mismatch contours) are used to construct contours of constant mismatch around a specified reflection coefficient. "MML" can be used to calculate the actual mismatch between a complex source and load. "S1P" and "S2P" (S11 prime and S22 prime) are used to calculate the input and output reflection coefficient with a specified load or source reflection coefficient, respectively.

Program Control

With the exception of the "CHR" and "STB" commands, the normal application of the functions in this section is interactive. That is, most information is used directly from the display without the requirement to print any output. However, the output can be printed from all the functions except the "CGC" and "MMC". These programs are in a loop waiting to calculate new values of constant gain or mismatch, and a subroutine to print the output cannot return program control to the ROM. However, the "LBL xx" (found in the Appendix beside these commands) can be used for this purpose. For example, the constant gain contours can be calculated, then a subroutine can be used to print the gain and location and radius of the circle. The subroutine can then be re-entered with a "GTO CGC", then "GTO xx" (GTO 29 in this case), and "R/S". The program will again ask for the desired gain. When this technique is used to print the output, you should note that the prompt "G=" or "L=" is in the alpha register and may be used as part of the printed output.

MAPPING

The mapping feature will map the parameters shown for two-port "A". The series and shunt mappings change the contents of two-port "A", as well as "B". These two-ports should be saved if you wish to use them following the mappings. The MAPSOURCE and MAPLOAD mappings produce a mapping of all possible source impedances on the S22 plane, and all possible load impedances on the S11 plane, respectively. These mappings are actually the S11' and S22' mappings, and are useful for determining the necessary reflection coefficient from a point on the constant gain contours. The transducer gain can then be used to verify the accuracy. The series and shunt feedback allows you to investigate the effects of feedback on any of the four S-parameters. All four S-parameters may be mapped from a single calculation for a specified feedback configuration; however, you must reload two-port "A" when a mapping for another feedback configuration is desired. Although the mappings are usually desired for the range of positive real impedances, you are free to specify negative real impedances if necessary. As with the "CGC" and "MMC" options, the mapping routine may be interrupted to print the output. The return is "GTO MAP", "GTO 16", and "R/S". The last prompt, "R=" or "X=" is in the alpha register and may be used with the print routine to label constant resistance or reactance.

Polar Coordinates

When using any of the mappings, including the contours described on pages 21 and 22. location and radius of the various circles are with respect to the chart used to plot the parameters. Thus, the center of any of the circles are found using the magnitude and angle from the chart, and measured from the center. The radius is set using the magnitude scale, and constructed around the location of the center. The Smith chart is normally used to plot the input and output reflection coefficients since impedance is usually desired; however, a polar plot is used for the transmission coefficients. In those cases where the purchased Smith Chart is not an appropriate size, such as might be the case with a large stability circle, the "chart" option of the "MAP" function may be used to construct a Smith Chart with any convenient scale factor.

EXAMPLES

The following examples will illustrate some of the applications and techniques used with HANDY-COMPACT. It is assumed that you are familiar with the design techniques used, as it is beyond the scope of this document to explain the theory and techniques behind the various examples. These examples all include the program listings and a detailed explanation of the circuit files used for the analysis. The listings also include most of the printer routines used to produce the output shown in each example.

Register Numbers

In the example illustrations, the numbers enclosed in circles next to the data correspond to the data register numbers where the parameter will be stored for analysis. This convention will be followed throughout these examples.

Printer Routines

You can write printer routines to suit your specific application, which can then be saved on a magnetic card and loaded as a subroutine when necessary, or deleted as required for more memory.

One 60-byte utility subroutine called "FMT" is not included with the listings in its final form. This subroutine formats data similar to the FORTRAN "F" format, and loads the formatted data into the print buffer via the "ACX"

command. When an overflow occurs, the specified fieldwidth is filled with "*"s. The call to this subroutine can be seen in the listings as the entry of a number, "xx.y" (fieldwidth.number of decimal places), then "XEQ FMT". The subroutine used here will fix the display to "y", then accumulate the number in the "y" register into the print buffer.

Example 1

Analysis of a Parallel-Coupled Filter

The four-section filter shown in Figure 1 has been designed for 0.25dB ripple in a passband which extends from 3.7 to 4.2 GHz, and should have a minimum of 25dB rejection at 4.5 GHz. The computed even- and odd-mode impedances for each coupled section is shown. The air length of 747.01601 mils for each coupled section is the same and is calculated from the requirement that each section must have an electrical length of 90 degrees at the center frequency of 3.95 GHz.




Figure 1. Parallel-Coupled Filter

The listing for the circuit file which is used for the analysis is shown in Figure 2. You must store the reference impedance in register 00, and the start, stop, and step frequencies in registers 20 through 22, respectively, before execution, as well as the data shown in the diagram.

01+LBL "EX1"	20 RCL 21
02 XEQ 00	21 RCL 22
03 RCL 35	22 RCL 20
04 RCL 36	23 +
05 RCL 37	24 STO 20
06 RCL 36	25 X<=Y?
07 XROM "CPL"	26 GTO "EX1"
08 XROM "EXC"	27 RTN
09 XEQ 00	28+LBL 00
10 RCL 40	29 1
11 RCL 36	30 STO 30
12 RCL 42	31 STO 32
13 RCL 36	32 0
14 XROM "CPL"	33 STO 31
15 XROM "CAS"	34 STO 33
16 XROM "EQU"	35 SF 03
17 XROM "CHA"	36 END
18 XROM "CAS"	
19 XEQ "PRD"	

Figure 2. Parallel-Coupled Filter Circuit File

The circuit file, which is labeled "EX1", makes use of the symmetry of this filter to reduce the execution time. Note in the above figure that the last two coupled line sections is a mirror image of the first two. Therefore, the first two sections can be cascaded with the mirror image to produce the overall parameters. The "CPL" element requires that the reflection coefficient for the two terminated ports be entered in registers 30 through 33, and a flag set which corresponds to the output port, port 3 in this case. The terminations and output ports are the same for all the coupled sections; therefore, a common subroutine, "LBL 00" located at steps 28 through 36, is used for this purpose. Program step 02 executes subroutine 00 which sets flag 3 and loads the termination of $1/\theta$ degrees (open circuit) into registers 30 through 33. Steps 03 through 06 recall the even-mode impedance and length, then the odd-mode impedance and length for the first section. Step 07 is the command to compute the S-parameters for the coupler. The S-parameters are stored as two-port "B" (registers 11 through 18). At this point there is no two-port "A" which can be connected; therefore, these parameters must be loaded into "A" for later connection. The exchange command, "EXC", in step 08 is used for this purpose. Steps 09 through 14 duplicate those above, but with the data for the second coupled section. Two-port "A", the first section, can now be cascaded with two-port "B", the second section, via the cascade, "CAS" command in step 15. Two-port "A" now contains the S-parameters for the first half of the filter. The equate command, "EQU", in step 16, copies the first half of the filter into "B", and the change command, "CHA", in step 17 exchanges ports 1 and 2 of two-port

"B" to form the mirror image of "A" in "B". The cascade connection in step 18 forms the S-parameters for the overall filter, then the print routine, "PRD", is called to print the output data. Step 20 through 22 recalls the stop, step, and current frequencies from registers 21, 22, and 20, respectively. Steps 23 and 24 adds the step to the current frequency and stores the result in register 20. A test is made in step 25 to determine if the current frequency is greater than the stop frequency. If it is, the program stops; otherwise the program loops to "EX1" to analyze the filter at the next frequency.

This program actually uses two printer subroutines, "PRL" and "PRD", which are shown in Figure 3.

01*LBL "PRL"		25 XEQ "FMT"
02 SF 12		26 2
03 CLA		27 SKPCHR
04 4		28 1
05 SKPCHR		29 RCL 01
06 ARCL 45		30 +
07 ACA		31 1
08 PRBUF	01*LBL "FMT"	32 RCL 01
09 CF 12	02 FRC	33 -
10 " FREQ"	03 10	34 /
11 ACA	04 *	35 7.2
12 4	05 FIX IND X	36 XEQ "FMT"
13 SKPCHR	06 RDN	37 2
14 "VSWR"	07 ACX	38 SKPCHR
15 ACA	08 END	39 RCL 03
16 5		40 X12
17 SKPCHR		41 LOG
18 "LOSS"		42 10
19 ACA		43 *
20 PRBUF		44 7.3
21 RTN		45 XEQ "FMT"
22*LBL "PRD"		46 PRBUF
23 RCL 20		47 END
24 6.2		

Figure 3. Two Printer Subroutines, "PRL" and "PRD"

"PRL" is used to print a label, and "PRD" is used to print data. Program "PRL" recalls the filename as alpha data from register 45 and prints this name along with the FREQ, VSWR, and LOSS headings. "PRL" is executed before execution of the circuit file, and "PRD" is called during execution to print the output data. Steps 23 through 25 recall the frequency and accumulate these data into the print buffer. A space of 2 character widths is accumulated; then steps 28 through 36 calculate and accumulate the input VSWR. The input VSWR is calculated from the relationship, $VSWR = (1+|S_{11}|)/(1-|S_{11}|)$. Two character widths are again accumulated in steps 37 and 38; then the insertion loss given by $10 \cdot \text{Log}[|S_{21}|^2]$ is calculated and accumulated in steps 39 through 45. Steps 46 and 47 print the print buffer and return to the circuit file.

The output shown in Figure 4, below indicates that the filter has a ripple of 0.316 dB instead of the design goal of 0.25dB. The rejection of 25dB at 4.5 GHz also has not been met. The analyzed value is 19.269dB. Three separate runs have been made. The first from 1 to 3 GHz with a step of 1 GHz, the second from 3.5 to 4.5 with a step of 0.1 GHz, and the last from 5 to 8 GHz with a step of 1 GHz. In this case, the even- and odd-mode impedances should be recalculated with more accuracy in order to achieve the desired 0.25dB. However, when the ripple is decreased, the rejection at 4.5 GHz will be even less which indicates that a five section filter must be used.

The five section must be analyzed in a slightly different manner since the symmetry about the center is not present. The first two sections should be cascaded and saved using the "SVA" command. The center section is created, then cascaded with the first two sections. The saved data is reloaded via the "USB" command into two-port "B". The "CHA" command is again used to generate the image of the first two sections and cascaded to form the overall parameters.

EX 1			4.00	1.31	-0.079
FREQ	VSWR	LOSS	4.10	1.54	-0.199
1.00	*****	-68.641	4.20	1.72	-0.316
2.00	*****	-55.125	4.30	13.87	-6.004
3.00	*****	-34.725	4.40	84.55	-13.353
3.50	84.55	-13.353	4.50	336.07	-19.269
3.60	13.87	-6.004	5.00	*****	-37.512
3.70	1.72	-0.316	6.00	*****	-56.616
3.80	1.54	-0.199	7.00	*****	-69.995
3.90	1.31	-0.079	8.00	*****	-90.892

Figure 4. Parallel-Coupled Filter Analysis

Another technique which may be used is to use the equivalent two-port model for coupled lines (see Matthaei, Young, and Jones, "Microwave Filters, Impedance-Matching Networks, and Coupling Structures"). The ladder analysis mode may then be used. The model for each coupled section will consist of three transmission lines, and since the ladder analysis is much faster than the two-port mode, execution time can be saved in some cases. However, in this case, the two-port mode is the preferred technique because of the symmetry.

Example 2

Microstrip Coupled Line Filter

Most filters similar to the one in Example 1 are constructed using either microstrip or stripline techniques. The microstrip filter is simple to fabricate and often used; however, the even- and odd-mode phase velocities are not the same due to the field patterns, and will cause a modification of the calculated response. This effect can easily be seen in the parameters of a microstrip coupler where the isolation is much less than expected.

The model of the microstrip version of the filter in Example 1 is shown in Figure 5. The even- and odd-mode effective dielectric constants have been calculated for each section, as well as the fringing capacitance at the end of each microstrip line. The average of the even- and odd-mode effective dielectric constants have been used to determine the physical length of each section (the air length divided by the square root of the average value). The physical length is then converted to the equivalent air length when multiplied by the square root of the appropriate dielectric constant.




Figure 5. Microstrip Coupled Line Filter

The circuit file, "EX2", shown in Figure 6, is used for the analysis of this structure.

```

01*LBL "EX2"          17 XROM "CAS"        33 STO 25
02 RCL 39             18 XROM "EQU"        34 STO 26
03 XEQ 00            19 XROM "CHA"        35 RDN
04 RCL 35            20 XROM "CAS"        36 XROM "CAP"
05 RCL 36            21 XEQ "PRD"        37 CF 00
06 RCL 37            22 RCL 21          38 X<Y
07 RCL 38            23 RCL 22          39 XROM "Z-R"
08 XROM "CPL"        24 RCL 20          40 STO 30
09 XROM "EXC"        25 +                41 STO 32
10 RCL 44            26 STO 20          42 RDN
11 XEQ 00            27 X<=Y?          43 STO 31
12 RCL 40            28 GTO "EX2"        44 STO 33
13 RCL 41            29 RTN                45 SF 03
14 RCL 42            30*LBL 00          46 END
15 RCL 43            31 SF 00
16 XROM "CPL"        32 0

```

Figure 6. Microstrip Coupled Line Filter Circuit File

Label 00 is used similar to Example 1 to set flag 3 and load the coupler terminations into registers 30 through 33. However, in this case, the reflection coefficient due to the fringing capacitance is desired rather than an open. This subroutine is called with the value of capacitance in register "x". Ladder analysis is by far the most efficient method for calculating the reflection coefficient, and the set flag 00 in step 31 sets this mode. The zero in step 32 is stored in registers 25 and 26 which result in a 0 ohm at 0 degree termination. The value of capacitance is returned to the "x" register in step 35, and step 36 calculates the input impedance of a series capacitor terminated with

a 0 ohm load. Step 37 returns the program to the two-port mode, and 38 interchanges the "x" and "y" registers for the conversion from impedance to reflection coefficient. You should be careful about this step; the conversion routine is set up for the normal entry of magnitude and angle for the impedance, and the output from the ladder analysis displays magnitude in "x" and angle in "y" which requires the interchange. Steps 40 through 46 then store the reflection coefficients, set flag 3, and return control to the circuit file.

The only other differences between the circuit files for Examples 2 and 3 are that the fringing capacitance which must be in "x" when subroutine 00 is called, and that the even- and odd-mode lengths are called from different data registers since they are not the same. The results of the analysis is shown in Figure 7.

EX2			4.00	1.47	-0.158
FREQ	VSWR	LOSS	4.10	2.45	-0.848
1.00	*****	-77.633	4.20	16.63	-6.695
2.00	*****	-61.469	4.30	75.53	-12.875
3.00	*****	-35.495	4.40	235.51	-17.736
3.50	13.81	-5.990	4.50	576.06	-21.599
3.60	1.44	-0.142	5.00	*****	-33.195
3.70	1.45	-0.149	6.00	*****	-40.269
3.80	1.00	0.000	7.00	*****	-31.777
3.90	1.65	-0.270	8.00	31.04	-9.174

Figure 7. Microstrip Coupled Line Filter Analysis

This analysis indicates that the passband is shifted to a lower frequency, the rejection on the low side has been improved, and the rejection on the high side has been dramatically decreased. If you have designed, built, and tested a parallel, coupled microstrip filter, you probably know that this is exactly the measured response. The question is whether the fringing capacitance or the unequal phase velocities cause the problem. In practice, the physical length of each section is reduced slightly to compensate for the capacitance; however, an analysis of this filter with the fringing set to zero shows that the phase velocity effect dominates. The filter should be redesigned, perhaps with an additional section, to achieve the desired passband and rejection.

Example 3

Two-Port Analysis of a Transformer Structure

The transformer structure shown in Figure 8 has been designed to transform a 100 ohm load into 50 ohms across a frequency band which extends from 2 to 8 GHz.




Figure 8. Transformer Structure

The design ripple is 0.05dB, and the minimum insertion loss is 0.001dB. Thus, the response should ripple between 0.001 and 0.051dB in this passband. The easiest method to analyze this structure is to incorporate the ideal 100:50 ohm transformer at the output port as shown above. This transformer will provide the desired 100-ohm termination for this structure, and will actually change the reference impedance at port 2. Thus, the calculated data are actually the S-parameters for a two-port with a reference impedance of 50 ohms at port 1, and 100 ohms at port 2. This approach will allow $|S_{21}|$ to be used directly to determine the insertion loss, and $|S_{11}|$ to be used for the VSWR calculation.

The circuit file for this analysis is shown in Figure 9.

01*LBL "EX3"	11 RCL 37	21 RCL 21
02 RCL 35	12 RCL 38	22 RCL 22
03 RCL 38	13 XROM "TRL"	23 RCL 20
04 SF 01	14 XROM "CAS"	24 +
05 XROM "SST"	15 100	25 STO 20
06 XROM "EXC"	16 ENTER↑	26 X<=Y?
07 RCL 36	17 50	27 GTO "EX3"
08 RCL 38	18 XROM "TRF"	28 END
09 XROM "TRL"	19 XROM "CAS"	
10 XROM "CAS"	20 XEQ "PRD"	

Figure 9. Transformer Structure
Circuit File (Two-Port)

Steps 02 and 03 recall the impedance and length of the input shunt shorted stub, and the set flag 01 in step 04 specifies that this element will be connected in shunt. The "SST" command in step 05 computes the S-parameters as two-port "B". The "EXC" command in step 06 places the parameters for the stub in "A" since this is the first element. Steps 07 through 10 recall the data for the first transmission line, then cascade this two-port with the shunt stub. Steps 11 through 14 do the same for the second transmission line. Steps 15 through 17 recall the impedance ratio for the transformer, and step 18 computes "B" as the parameters for this element. Note that the input impedance for the transformer can be stored in a data register, and the reference impedance in register 00 used for the output impedance. The "CAS" in step 18 computes the overall S-parameters, and the call to "PRD" computes and prints the results.

The output is shown in Figure 10.

EX3		
FREQ	VSWR	LOSS
1.00	3.96	-1.914
2.00	1.20	-0.034
3.00	1.27	-0.060
4.00	1.25	-0.053
5.00	1.02	0.000
6.00	1.25	-0.053
7.00	1.27	-0.060
8.00	1.20	-0.034
9.00	3.96	-1.914

Figure 10. Transformer Structure Two-Port Analysis

The analysis indicates that the design values are not exact. The response at the center of the band should be 0.001dB, and at both band edges, 0.051dB. However, the results do show that this network would be acceptable for most applications.

Example 4

Ladder Analysis of a Transformer Structure

The Transformer Structure shown in Figure 8 can be analyzed using ladder analysis. The circuit file is shown in Figure 11.

```
01+LBL "EX4"          13 RCL 35            25 STO 03
02 SF 00              14 RCL 38            26 XEQ "PRD"
03 100                15 SF 01             27 RCL 21
04 STO 25             16 XROM "SST"        28 RCL 22
05 0                  17 X<>Y              29 RCL 20
06 STO 26             18 XROM "Z-R"        30 +
07 RCL 37             19 STO 01            31 STO 20
08 RCL 38             20 X↑2               32 X<=Y?
09 XROM "TRL"         21 CHS               33 GTO "EX4"
10 RCL 36             22 1                 34 CF 00
11 RCL 38             23 +                 35 END
12 XROM "TRL"         24 SQRT
```

Figure 11. Transformer Structure Circuit File (Ladder)

The ladder method will result in a much faster analysis, and should be almost as accurate as the two-port method.

Ladder analysis can only provide the input impedance to the network; however, the impedance may be used to directly compute the input VSWR, and a relationship for lossless two-ports may be used to compute the loss. $|S_{21}|^2$ for a lossless two-port can be calculated from: $1 - |S_{11}|^2$. Thus, the input impedance can be used to

determine the insertion loss as well. Flag 00 is set in step 02 to specify ladder analysis. The magnitude of the terminating impedance (100 ohms) is stored in register 25, and the angle (0 degrees) in register 26 in steps 03 through 06. You should be aware that these steps must be within the frequency loop since at the conclusion of the analysis these registers will contain the input impedance, and must be initialized to the terminating impedance. Steps 07 through 16 are similar to Example 3 above, except the structure is analyzed from the output to the input, and it is not necessary to cascade the elements. Steps 17 and 18 exchange the magnitude and angle of the impedance in "x" and "y" for the conversion to reflection coefficient in step 18. The magnitude of the reflection coefficient is stored in register 01, and the magnitude of S21 calculated and stored in register 03 in steps 19 through 25. The call to "PRD" in step 26 prints the results, the frequency is incremented, and the stop test is made. When the analysis is complete, step 34 clears flag 00 (resetting the two-port mode) and stops. The results of the analysis are identical to those shown in Figure 10.

In most cases the ladder analysis will agree with the two-port analysis. However, when the insertion loss is high, the agreement will not be good and the two-port analysis should be used. This problem is caused when $|S11|$ is squared and subtracted from one to calculate $|S21|^2$. When the insertion loss is high, $|S11|$ is almost one, and the accuracy is lost when it is squared and subtracted from one. This problem starts to occur somewhere near 60 to 70dB. It may be worthwhile in a very complicated analysis

to have two circuit files. The ladder analysis can be used unless the insertion loss is greater than some preset value, in which case the analysis would shift to the two-port mode until the insertion loss again rises above the test value.

The network was analyzed using a 100-ohm termination as the load. However, there is no reason why the network could not be computed in reverse order. In this case, the termination would be 50 ohms, and the network analyzed starting with the shorted stub. The reference impedance in register 00 should be changed to 100 ohms so that the reflection coefficient is referenced to this value. Two-port analysis can also be performed in this manner with a 100-ohm reference. The network would be analyzed in the reverse order, and the transformer would transform from 100 to 50 ohms. You should also note that $|S_{11}|$ and $|S_{21}|$ was calculated and stored in registers 01 and 03 for a call to "PRD", which uses these data to compute the VSWR and insertion loss. These data can be computed and printed as a part of the circuit file without a call to an external file. However, "PRD" is used by several of these examples and represents a common subroutine.

Example 5

Plotting Network Response

The plotting capability of the printer can be used to plot the network response as well as to print the output. The requirement of these built-in routines is that the function be called with the value x in the "x" register, and the value of y , $f(x)$, be returned in the "x" register. The plotting routines also use registers 00 through 11 which conflicts with HANDY-COMPACT, although there is a way around this problem. The listings in Figure 12 show the circuit file for "EX5" and a plot routine named "PLOT".

01*LBL "EX5"	38*LBL 00
02 STO 20	39 RCL IND 52
03 XEQ "EXR"	40 X<> IND 53
04 SF 00	41 STO IND 52
05 100	42 ISG 52
06 STO 25	43 ISG 53
07 0	44 GTO 00
08 STO 26	45 END
09 RCL 37	
10 RCL 38	
11 XROM "TRL"	
12 RCL 36	01*LBL "PLOT"
13 RCL 38	02 "X MIN ?"
14 XROM "TRL"	03 PROMPT
15 RCL 35	04 STO 00
16 RCL 38	05 "X MAX ?"
17 SF 01	06 PROMPT
18 XROM "SST"	07 STO 09
19 X<>Y	08 "X INC ?"
20 XROM "Z-R"	09 PROMPT
21 CF 00	10 STO 10
22 X+2	11 FIX IND 05
23 CHS	12*LBL 00
24 1	13 RCL 08
25 +	14 ACX
26 LOG	15 3
27 10	16 SKPCOL
28 *	17 RCL 08
29 STO L	18 XEQ IND 11
30 XEQ "EXR"	19 REGPLOT
31 RCL L	20 RCL 10
32 RTN	21 ST+ 00
33*LBL "EXR"	22 RCL 09
34 .012	23 RCL 08
35 STO 52	24 X<=Y?
36 40.051	25 GTO 00
37 STO 53	26 END

Figure 12. Circuit File with Plotting Routine

The circuit file is the same as "EX4" above with some minor changes. This file will be called from the plot routines with the value of x (the frequency) in the "x" register. Step 02 stores this value in frequency register 20. Since the plot routines use registers 00 through 11 and cause a conflict with the ladder analysis, a subroutine named "EXR" is called from step 03. This subroutine found in steps 33 through 45 exchange the contents of registers 00 through 11 with the contents of registers 40 through 51. The remainder of the circuit file from steps 04 through 20 is identical to that portion of Example 4; in fact, a modified version of "EX4" could be called at this point. Step 21 clears flag 00, and steps 22 through 29 calculate the insertion loss in dB and store this value in register "L" for return to the plot routine. The data registers are again exchanged, the insertion loss recalled from "L", and execution returned to the plot routine to plot this point.

A review of the technique which must be used in the plot mode is:

1. Store the frequency from the "x" register into register 20,
2. Exchange registers 00 through 11 with registers not used by the circuit file,
3. Analyze the circuit and compute the parameter which is to be plotted, and
4. Save the parameter, restore registers 00 through 11, and return to the plot routine with the plot parameter in register "x".

The results are shown in Figure 13, below.




Figure 13. Network Response Plot

The initial analysis from 1 to 2 GHz was performed using the standard "PRPLOT" function of the printer. Note that the reference impedance must either be stored in register 40, or stored in register 00 and "EXR" executed from the keyboard since registers 40 through 51 will be loaded into 00 through 11 before the analysis. The plot is desired with a change in the scale for the x parameter (frequency); it is desired to plot the upper and lower skirts in steps 0.25 GHz, and the passband in steps of 2 GHz. Neither the "PRPLOT" or the "REGPLOT" routines have provisions to continue a plot and

label the x axis. The "PLOT" routine is used for this purpose. The "PLOT" routine shown above prompts for the minimum, maximum and step value for the x variable, then uses the data already stored by the "PRPLOT" function to accumulate the label, and executes a call to "REGPLOT" for the actual plot. The program also increments x, and tests for a stop. The plot shown above was made with one call to "PRPLOT" and two calls to "PLOT".

Example 6

Analysis of a Feedback Amplifier

The feedback amplifier, shown in Figure 14, has been designed to provide 10dB gain from 250 to 1000 MHz. This structure will be analyzed from 250 to 1000 MHz in 250 MHz steps.




Figure 14. Feedback Amplifier

The S-parameters are shown in Figure 15 in the order of increasing frequency.

**TWO-PORT A
S-PARAMETERS**

S11= 0.50
 -75.00
 S12= 0.02
 65.00
 S21= 21.00 0.25 GHz
 145.00
 S22= 0.70
 -18.00

**TWO-PORT A
S-PARAMETERS**

S11= 0.48
 -153.00
 S12= 0.04
 45.00 0.75 GHz
 S21= 6.20
 109.00
 S22= 0.49
 -29.00

**TWO-PORT A
S-PARAMETERS**

S11= 0.58
 -120.00
 S12= 0.04
 40.00
 S21= 9.40 0.50 GHz
 121.00
 S22= 0.52
 -28.00

**TWO-PORT A
S-PARAMETERS**

S11= 0.58
 -144.00
 S12= 0.05
 49.00 1.00 GHz
 S21= 4.50
 102.00
 S22= 0.46
 -30.00

Figure 15. Feedback Amplifier S-Parameters

These parameters may be saved in data memory and recalled during the analysis. This approach will require that the calculator be sized for a minimum of 067 (see the Appendix under the "SVA" command) plus the required storage for the element values. Actually, one additional two-port must be saved during the analysis, which will require a size of 075 for the two-port data. However, if you do not have sufficient memory, these parameters may be entered at execution time, either directly from the

keyboard, or using magnetic cards. The temporary data may be saved on a data card as well, or these values recorded for reentry from the keyboard. If you elect to use the latter technique, you can use the "PROMPT" command to prompt for data entries, reducing the possibility of entering data for the wrong frequency.

The circuit file used for the analysis is shown in Figure 16 below.

01+LBL "EX6"	32 5	63 *
02 1	33 XROM "USA"	64 6.2
03 STO 34	34 XROM "CAS"	65 XEQ "FMT"
04+LBL 00	35 SF 01	66 PRBUF
05 SF 01	36 RCL 82	67 "ISWR="
06 RCL 75	37 XROM "RES"	68 ACA
07 XROM "CAP"	38 XROM "CAS"	69 1
08 XROM "EXC"	39 XEQ "PR"	70 RCL 01
09 RCL 76	40 RCL 21	71 +
10 XROM "IND"	41 RCL 22	72 1
11 XROM "CAS"	42 RCL 20	73 RCL 01
12 SF 01	43 +	74 -
13 RCL 77	44 STO 20	75 /
14 XROM "RES"	45 X<=Y?	76 6.2
15 XROM "CAS"	46 GTO 00	77 XEQ "FMT"
16 5	47 RTN	78 2
17 XROM "SVA"	48+LBL "PR"	79 SKPCHR
18 RCL 34	49 FIX 2	80 "OSWR="
19 XROM "USA"	50 "FREQ="	81 ACA
20 1	51 ACA	82 1
21 ST+ 34	52 RCL 20	83 RCL 04
22 SF 01	53 6.2	84 +
23 RCL 80	54 XEQ "FMT"	85 1
24 RCL 81	55 2	86 RCL 04
25 XROM "PRC"	56 SKPCHR	87 -
26 XROM "SER"	57 "GAIN="	88 /
27 RCL 78	58 ACA	89 6.2
28 RCL 79	59 RCL 03	90 XEQ "FMT"
29 XROM "SRL"	60 X12	91 PRBUF
30 XROM "PAR"	61 LOG	92 ADV
31 XROM "EXC"	62 10	93 END

Figure 16. Feedback Amplifier Circuit File

This amplifier uses both series and shunt feedback requiring the use of temporary storage for one two-port. Recall that only two two-ports may be active at any time, and the connection of the feedback elements with the device will require both. During this process, the parameters for the input network must be saved. The procedure used for this analysis will be to analyze the input network, save the data, connect the series and shunt feedback elements, reload and cascade the input network, then connect the output network. However, there is no reason why the feedback elements could not be connected with the device, then saved while the input network is cascaded.

The index register, register 34, will be used to recall the device S-parameters at the corresponding frequencies. Before execution, these parameters should be entered from the keyboard (or loaded from data cards) and saved in two-port 1 through two-port 4 using the "SVA" command. Two-port 5 will be used as temporary storage during the analysis. Steps 02 and 03 in the circuit file initialize the index register to 1. Label 00 in step 04 will be used as the frequency loop. This label must be used since the index register will be incremented at each frequency, and should not be reset to 1. Some program memory can be saved by loading the index register from the keyboard before execution; label 00 may be deleted as well (the GTO in step 46 must also be changed), but you must remember this step each time the circuit is analyzed. Steps 05 through 07 compute the S-parameters of the input shunt capacitor as two-port "B". The "EXC" command in step 08 saves the parameters in "A" since this is the first element. Step 09

through 11 computes the series inductor and cascades it with the capacitor, and steps 12 through 15 do the same for the shunt resistor. At this point, the entire input network is in two-port "A", and the commands in steps 16 and 17 save this data in two-port 5 (registers 67 through 74). The index register (1 at the first frequency) is recalled in step 18, then the corresponding two-port is recalled into "A" by step 19. Steps 20 and 21 increment the index register for the next frequency. Note that the calculator command "ISG" can be used if the index register is loaded with 1 plus a fraction greater than .004, (1.004) may be entered and "ISG" used as the stop test. In this case, the frequency should be incremented before the test, and the stop frequency need not be used.

Steps 22 through 25 recall the data for the series feedback element, set the shunt flag, and calculate the S-parameters for this element. The "SER" command in step 26 connects this element with the device. Steps 27 through 30 do the same for the shunt feedback which is connected using the "PAR" command. Two-port "A" contains the parameters for the device with both series and shunt feedback, and the "EXC" command in the next step loads this data into "B". Steps 32 through 34 recall the input network and cascade it with the device, then the output element is calculated and cascaded in the next four steps to form the overall amplifier parameters. The subroutine "PR" is called to print the output data, the frequency is incremented, a stop test is then made in the usual manner.

The output routine in Figure 17 prints the frequency, then computes the gain from $|S_{21}|$, the input VSWR from $|S_{11}|$, and the output VSWR from $|S_{22}|$. The output indicates that this amplifier does indeed have 10dB gain from 250 to 1000 MHz. Maximum VSWR is 1.74:1, confirming that the design is correct.

EX6

FREQ= 0.25 GAIN= 9.97
ISWR= 1.19 OSMR= 1.21

FREQ= 0.50 GAIN= 9.82
ISWR= 1.51 OSMR= 1.40

FREQ= 0.75 GAIN= 10.01
ISWR= 1.71 OSMR= 1.59

FREQ= 1.00 GAIN= 10.04
ISWR= 1.74 OSMR= 1.61

Figure 17. Feedback Amplifier Analysis

Two utility programs, "PTA" and "PTB" in Figure 18, can be used to print the S-parameters of either two-port "A" or "B", respectively.

01+LBL "PTA"	22 "S21"
02 1	23 XEQ 01
03 "A"	24 "S22"
04 GTO 00	25+LBL 01
05+LBL "PTB"	26 "+= "
06 11	27 ACA
07 "B"	28 RCL IND X
08+LBL 00	29 ACX
09 ADV	30 PRBUF
10 ASTO L	31 5
11 " TWO-PORT "	32 SKPCHR
12 ARCL L	33 RDN
13 SF 12	34 4
14 PRA	35 ST+ Z
15 "S-PARAMETERS"	36 RCL IND Z
16 PRA	37 ACX
17 CF 12	38 PRBUF
18 "S11"	39 R↑
19 XEQ 01	40 3
20 "S12"	41 -
21 XEQ 01	42 END

Figure 18. Print Utility Programs

"PTA" was used to print the device parameters. These subroutines do not print the frequency as shown. However, it is suggested that several subroutines be written to print the various parameters, input VSWR, output VSWR, gain, frequency, etc., with the desired headings. These may then be called as desired. For example, subroutine "FRQ" may be called then subroutine "GAN" called to accumulate the frequency and gain similar to subroutine "PR" above. These routines may then be combined to produce the desired output.

Example 7

Analysis of a Matching Network

The matching network in Figure 19 has been designed to match a complex load into 50 ohms across the 1.0 to 2.0 GHz band. The load impedance has been measured, and could represent an antenna impedance or the input to an active device.




Figure 19. Matching Network

The impedance is given in the form as shown in the form:

<u>FREQ</u>	<u>IMPEDANCE</u>			
	<u>Real</u>	<u>register</u>	<u>Imag</u>	<u>register</u>
1.0	1.9	48	+j 2.0 ohms	49
1.2	2.0	50	+j 2.1 ohms	51
1.4	2.0	52	+j 2.0 ohms	53
1.6	1.9	54	+j 1.8 ohms	55
1.8	1.8	56	+j 1.5 ohms	57
2.0	1.7	58	+j 1.0 ohms	59

These data are stored in the indicated registers, However, they are stored in polar form, with magnitude in the lower numbered register and angle in the next highest. The ladder analysis will be used, requiring the impedance in polar form. These data may also be stored in rectangular form and converted when recalled. The circuit file shown below is similar to Example 4 with some minor exceptions. The termination impedance is complex and must be recalled at each corresponding frequency. Steps 03 and 04 initialize the index register to the location of the magnitude for the first frequency. Step 08 recalls the magnitude and stores it into the ladder analysis impedance magnitude register. Steps 10 through 13 increment the index register, then recall and store the angle of the measured impedance. The index register is then incremented for use at the next frequency in steps 14 and 15. Steps 16 through 32 connect each element of the matching network, and convert the input impedance to reflection coefficient. Steps 33 through 64 save the input reflection coefficient, accumulate the frequency, then compute and print the input VSWR. The input reflection coefficient is recalled and printed, the frequency incremented, and the stop test made.

As shown in the program listing (Figure 20), the program does not loop back to the beginning of the circuit file so that the index register is unchanged.

01*LBL "EX7"	26 RCL 63	51 6
02 SF 00	27 XROM "TRL"	52 ACCHR
03 48	28 RCL 60	53 "= "
04 STO 34	29 RCL 61	54 ACA
05 1	30 XROM "TRL"	55 RCL 68
06 STO 20	31 X<>Y	56 8.5
07*LBL 15	32 XROM "Z-R"	57 XEQ "FMT"
08 RCL IND 34	33 STO 68	58 6
09 STO 25	34 X<>Y	59 SKPCHR
10 1	35 STO 69	60 RCL 69
11 ST+ 34	36 RCL 20	61 7.2
12 RCL IND 34	37 4.1	62 XEQ "FMT"
13 STO 26	38 XEQ "FMT"	63 PRBUF
14 1	39 10	64 ADV
15 ST+ 34	40 SKPCHR	65 RCL 21
16 RCL 67	41 1	66 RCL 22
17 SF 01	42 RCL 68	67 RCL 20
18 XROM "CAP"	43 +	68 +
19 RCL 65	44 1	69 STO 20
20 RCL 66	45 RCL 68	70 X<=Y?
21 XROM "TRL"	46 -	71 GTO 15
22 SF 01	47 /	72 CF 00
23 RCL 64	48 6.3	73 END
24 XROM "CAP"	49 XEQ "FMT"	
25 RCL 62	50 PRBUF	

Figure 20. Matching Network Circuit File

Again, the index register may be initialized from the keyboard before program execution, but you must be careful to remember this step. The

result of this analysis is shown in the listing in Figure 21. This matching network is not acceptable since the input VSWR ranges from 2.444 to 3.972:1. The matching network should either be redesigned with more care, or a CAD program with the capability of optimization should be used to reduce the VSWR to an acceptable level.

EX7

FREQ	VSWR
1.0	2.444
$\Gamma = 0.41928$	-53.23
1.2	2.702
$\Gamma = 0.45977$	-98.77
1.4	3.525
$\Gamma = 0.55800$	-140.27
1.6	3.972
$\Gamma = 0.59774$	-179.26
1.8	3.626
$\Gamma = 0.56764$	139.20
2.0	2.637
$\Gamma = 0.45009$	84.84

Figure 21. Matching Network Analysis

Example 8

Amplifier Design

HANDY-COMPACT can be used as a design tool. In this example, it will be used in the initial design phase to set design goals and define the required impedances for a matching network. The network may then be modeled and analyzed or optimized as desired. The design capability can be illustrated through the partial design of an amplifier.

The goal will be to design a 25dB gain amplifier at 8.0 GHz. (The technique could be extended to the design of amplifiers with wider bandwidths.) The first task is to characterize the device so that the number of stages and gain per stage can be determined. The characterization ("CHR") and stability ("STB") functions can be used to determine the necessary information.

The S-parameters of a commercially available GaAs FET are shown below:

<u>FREQ</u>	<u>S11</u>	<u>S12</u>	<u>S21</u>	<u>S22</u>
8.0	0.717 -57	0.028 171	2.223 138	0.851 -20
10.0	0.674 -75	0.041 125	1.983 125	0.762 -21
12.0	0.602 -98	0.056 156	1.866 113	0.611 -30
14.0	0.582-121	0.077 139	1.495 87	0.761 -79

Both ROM programs are modified to include the automatic execution feature, as well as format and add headings to the output. Both routines are shown in Figure 22, below. These programs have been implemented by programming the file as shown with the exception of the labels found in step 01. This approach allows the calculator to convert the call to the ROM programs with the same name to "XROM" so that the ROM program will execute. The last step of the programming for each file is to go to step 00, and add the "LBL" "name". Thus when the program is executed, the program in RAM, which uses the program with the same name in ROM as a subroutine, will execute as desired. (This technique is described on Page 10.)

01*LBL "CHR"	41 SF 13	06 XEQ 00
02 SF 04	42 "A="	07 2
03 XROM "CHR"	43 ACA	08 XROM "STB"
04 CF 04	44 RCL 09	09 CF 04
05 ADV	45 9.5	10 "OUTPUT"
06 3	46 XEQ "FMT"	11*LBL 00
07 SKPCHR	47 PRBUF	12 "STABILITY"
08 "FREQ= "	48 6	13 4
09 ACA	49 ACCHR	14 SKPCHR
10 RCL 20	50 "S="	15 ACA
11 8.4	51 ACA	16 PRBUF
12 XEQ "FMT"	52 RCL 24	17 "CTR="
13 " GH"	53 8.5	18 ACA
14 ACA	54 XEQ "FMT"	19 RCL 31
15 SF 13	55 3	20 8.5
16 "Z"	56 SKPCHR	21 XEQ "FMT"
17 ACA	57 RCL 27	22 2
18 CF 13	58 10.5	23 SKPCHR
19 PRBUF	59 XEQ "FMT"	24 RCL 32
20 "K="	60 PRBUF	25 10.5
21 ACA	61 6	26 XEQ "FMT"
22 RCL 29	62 ACCHR	27 PRBUF
23 7.4	63 "L="	28 "RAD="
24 XEQ "FMT"	64 ACA	29 ACA
25 3	65 RCL 19	30 RCL 33
26 SKPCHR	66 8.5	31 8.5
27 1	67 XEQ "FMT"	32 XEQ "FMT"
28 RCL 29	68 3	33 2
29 X>Y?	69 SKPCHR	34 SKPCHR
30 GTO 00	70 RCL 23	35 "STABLE "
31 "MSG"	71 10.5	36 1
32 ACA	72 XEQ "FMT"	37 RCL 10
33 RCL 09	73 PRBUF	38 X=0?
34 9.5	74 CF 13	39 "FOUT"
35 XEQ "FMT"	75 END	40 X=Y?
36 PRBUF	01*LBL "STB"	41 "FIN"
37 RTN	02 SF 04	42 ACA
38*LBL 00	03 1	43 PRBUF
39 "G"	04 XROM "STB"	44 END
40 ACA	05 "INPUT"	

Figure 22. Amplifier Design Aids

The characterization program, "CHR", shown in Figure 22, sets the automatic execution flag (flag 04) and calls the ROM program which computes the stability factor, gain, and simultaneous matched source and load for a stable network. The ROM program returns to RAM with the parameters in the data registers indicated in the Appendix. The program clears the automatic execution flag, then prints the frequency, and accumulates the stability factor in steps 04 through 24. A test is made in steps 27 through 30 to determine whether the stability factor is greater than one. When the stability factor is greater than one, execution transfers to label 00 in step 38 where the maximum available gain as well as the simultaneous, matched load- and source-reflection coefficient is printed. Otherwise, execution continues where the maximum stable gain is printed.

The stability program, "STB" above, sets the automatic execution flag, and calls the ROM program twice: once with the code for input plane and again with the output plane code. Following each call to the program, the title "INPUT" or "OUTPUT" is loaded into the alpha register and a call is made to label 00 where the data is printed. Label 00 at step 11 appends "STABILITY" to the data already in the alpha register and prints the heading. The magnitude and center of the stability circle is then printed in steps 17 through 27. Steps 28 through 35 accumulate the radius and "STABLE" into the print buffer. Steps 36 through 41 test the stability code to determine the location of the stable region. If the network is not unconditionally stable, either "IN" or "OUT" is accumulated into the buffer and the data printed.

The results of this analysis is shown in Figure 23.

FREQ= 8.0000 GHz
K= 0.5592 MSG 18.99781
INPUT STABILITY
CTR= 1.19532 51.59367
RAD= 0.30189 STABLE OUT
OUTPUT STABILITY
CTR= 1.09068 17.53045
RAD= 0.14952 STABLE OUT

FREQ= 12.0000 GHz
K= 1.6704 Ga= 10.44397
 Γ_s = 0.72637 92.87296
 Γ_l = 0.73224 25.00021
INPUT STABILITY
CTR= 1.54898 92.87296
RAD= 0.37645 STABLE
OUTPUT STABILITY
CTR= 1.53010 25.00021
RAD= 0.36220 STABLE

FREQ= 10.0000 GHz
K= 0.9904 MSG 16.84539
INPUT STABILITY
CTR= 1.31247 70.24105
RAD= 0.31476 STABLE OUT
OUTPUT STABILITY
CTR= 1.20969 17.93536
RAD= 0.21136 STABLE OUT

FREQ= 14.0000 GHz
K= 1.0863 Ga= 11.09017
 Γ_s = 0.84652 105.00059
 Γ_l = 0.91271 72.41855
INPUT STABILITY
CTR= 1.70842 105.00059
RAD= 0.67403 STABLE
OUTPUT STABILITY
CTR= 1.29870 72.41855
RAD= 0.27996 STABLE

Figure 23. Amplifier Analysis

The data show that the device is not stable at 8.0 GHz and that the maximum stable gain is 18.99781dB. Therefore, two gain stages, each with 12.5dB gain, will be required for the design. Several approaches may be taken to achieve the desired gain, however, for purposes of this example it will be assumed that noise figure is a consideration, and a lossless reactive mismatch will be used. This approach will result in a mismatch at either the input or output, and will require some means to overcome this problem. The two methods most commonly used are the hybrid-coupled amplifier and the isolator. The isolator is a narrow-band device, and the impedances change very fast outside the passband. Since the device is unstable and out-of-band impedances will be critical to the stability, the balanced amplifier approach is used.

The source and load impedances must now be found which will result in the desired 12.5dB gain, and do not fall within the unstable regions. One approach which may be used is to generate the constant gain contours on either the source or load planes, select an impedance, then compute the conjugate match on the opposite plane. Either plane may be used for the gain circles, however, the output impedance is usually less sensitive to variations in the input impedance, and gain circles on the input plane will allow the selection of an impedance which will provide low noise as well as the specified gain. Figures 24 and 25 show the respective input and output planes for this device at 8.0 GHz.

NAME	TITLE INPUT PLANE - 8.0 GHz	DWG. NO.
SMITH CHART FORM 2Y-91-N	ANALOG INSTRUMENTS COMPANY, NEW PROVIDENCE, N.J. 07974	DATE

NORMALIZED IMPEDANCE AND ADMITTANCE COORDINATES




Figure 24. Device Input Plane at 8 GHz

NAME	TITLE OUTPUT PLANE - 8.0 GHz	DWG. NO.
SMITH CHART FORM ZY-91-N	ANALOG INSTRUMENTS COMPANY, NEW PROVIDENCE, N.J. 07974	DATE

NORMALIZED IMPEDANCE AND ADMITTANCE COORDINATES




Figure 25. Device Output Plane at 8 GHz

Both planes show the stability circles constructed from the output of the stability analysis. The shaded portion is unstable, and the impedances must not be selected from these regions. Constant gain contours for the input plane were calculated for several values of gain, and are shown on the input plane. The output data from the "CGC" program is shown below:

<u>GAIN</u>		<u>CENTER</u>	<u>RADIUS</u>
12.5	0.50912	51.59367	0.51335
15.0	0.67996	51.59367	0.36410
17.0	0.80862	51.59367	0.26921
19.5	0.94199	51.59367	0.21242

These data may be saved by interrupting the program and returning to the label in ROM as mentioned in the text above, or as in this case, by using the print stack ("PRSTK") command. The 12.5dB gain contour passes through the origin of the mapping, 50 ohms, and is a good candidate for the source impedance. No input matching network will be required. The corresponding output reflection coefficient, S_{22}' , calculated from "S2P" with $0/0$ degrees, is shown on the output plane as $S_{22}'A$, and is $0.85100/-20.0$ degrees. The load reflection coefficient is the complex conjugate of this value, or $0.85100/20.0$ degrees. The transducer gain, "GTP", can be calculated using these points, and is 12.53286dB.

It appears from the constant gain contours that a gain can be found which is higher than the

maximum stable value of 18.99781dB without causing the input reflection coefficient to enter the unstable region. Point "C" is such a point on the 19.5dB gain circle. This point is $0.74/51$ degrees. The corresponding output reflection is $0.94552/18.649$ degrees. This point does fall within the unstable region of the output stability circle, as shown. However, note that the reflection coefficient is less than one, thus positive real. The input reflection coefficient, S_{11}' , calculated using this load with the "S1P" program, is $1.00479/-51.40$ degrees which results in a negative real input impedance as expected.

Point "A" and the corresponding output load may be used to provide a simple 12.5dB gain stage. However, in the case of a broadband design, it is desirable to find a source and load reflection coefficient which results in the desired gain at each frequency, and follows either the constant resistance or conductance contours with frequency. This approach results in a simple load which consists of either a series or parallel combination of a resistor and a reactive element. There are many techniques available to match a load represented in this fashion. Therefore, it is desirable to be able to select a range of impedances in both planes. The mismatch contours, "MMC", can be used for this purpose. Point "B" is selected as the input impedance at this frequency, and is assumed to connect with the reflection coefficient at another frequency following the constant resistance or conductance contours. This point is $0.615/66$ degrees, and the corresponding S_{22}' , $0.90217/-17.140$ degrees, as shown on the output plane. Since 12.5dB is the

desired gain, and point "B" provides 17dB gain, a 4.5dB mismatch will be required on the output plane. The "MMC" program is used with S22' as the input to calculate the 4.5dB mismatch contour. The center is located at $0.67408/17.140$ degrees, and has a radius of 0.31476. This mismatch contour is shown on the output plane, and any load impedance selected from this contour will result in the desired 12.5dB gain. The point on this contour is selected for the same reasons as the input, and is shown as Γ_{LB} , which is $0.48/44$ degrees. Thus, the source reflection coefficient is $0.615/66$ degrees, and the load, $0.48/44$ degrees. These numbers should be checked before the design continues to check for possible errors. The transducer gain, "GTP", calculated from these points is 12.46131dB (which is acceptable), and the design can continue.

The matching networks designed directly from the Smith chart are shown in Figure 26.




Figure 26. Input Matching Network

The input network is a simple two-element network which transforms 50 ohms into the desired source impedance.

The output network uses a quarter-wavelength transmission line to transform the real portion, and a shunt inductor to provide the desired output.

The ladder analysis mode is set, and the reflection coefficient presented by each network is calculated directly from the keyboard as a check. The input reflection is $0.59484/66.46487$ degrees, and the output, $0.47868/43.18923$ degrees. The transducer gain corresponding to these terminations is 12.4510dB. The program is returned to the two-port mode, and a complete analysis is then made directly from the keyboard.

The results are shown in Figure 27, and the gain calculated from $|S_{21}|$ is 12.45103dB, almost exactly that predicted by the ladder analysis. Characterization of this network reveals that the stability factor and maximum stable gain is exactly that calculated for the device alone, as it should be using lossless matching networks.

```
TWO-PORT A
S-PARAMETERS
S11= 0.35413
      -13.94719
S12= 0.05282
      113.99843
S21= 4.19326
      80.99843
S22= 0.80001
      -128.84733
```

Figure 27. Results of Complete Analysis

Since it is assumed that this amplifier will be balanced, the final step in the design is the analysis of the overall stage. Assume that the amplifier is an 8 to 12 GHz design, which would have a coupler center frequency of 10 GHz, and would be over-coupled at the center frequency. A coupling coefficient of 2.5dB will be assumed for this analysis. The analysis of two identical stages is accomplished, again from the keyboard, by executing the "EQU" command which duplicates two-port "A" in "B", then entering the coupler parameters and executing the hybrid connection, "HYB". The results are shown in Figure 28.

```
TWO-PORT A
S-PARAMETERS
S11= 0.02656
      10.31544
S12= 0.05267
      48.26106
S21= 4.18144
      15.26106
S22= 0.06001
      -104.58470
```

Figure 28. Results of Hybrid Connection Analysis

The overall balanced amplifier has an input VSWR of 1.05:1, output VSWR of 1.13:1, gain of 12.43dB, reverse isolation of -25.57dB, and is unconditionally stable (the stability factor is 2.37). The final steps in the design should be to analyze the structure across a broad frequency range, especially at the lower frequencies where the device becomes even more unstable.

The approach taken above is an extremely fast and effective method to define the initial design parameters, and provides sufficient accuracy even using graphical techniques. HANDY-COMPACT provides the necessary data to allow flexibility in the design tradeoffs. Another useful technique is to use the source and load mappings of "MAP". These mappings provide immediate information as to the required source or load impedance which corresponds to the termination in the opposite plane. This approach allows an impedance to be selected from one plane, which results in a desirable impedance in the other. Stability information is immediately available without the requirement for stability circles.

APPENDIX

This Appendix serves as a reference guide to HANDY-COMPACT. The Tables in this Appendix correspond (to some extent) with the text discussions at the front of the manual:

<u>Table</u>	<u>Page</u>
PROGRAM STRUCTURE	85
ELEMENT LIBRARY	86-87
INTERCONNECTIONS	88
LADDER ANALYSIS	89
MANIPULATION COMMANDS	90
CONVERSION COMMANDS	91
CHARACTERIZATION	92-93
MAPPING	94

SIZING

To size the calculator, enter:

XEQ ALPHA SIZE ALPHA 035

STRUCTURE

SIZE = 035	
PROGRAM UNITS: GHZ OHMS MMHO NH PF MILS Er = 1	
$S_A = \begin{bmatrix} \textcircled{1} / \textcircled{5} & \textcircled{2} / \textcircled{6} \\ \textcircled{3} / \textcircled{7} & \textcircled{4} / \textcircled{8} \end{bmatrix}$	
$S_B = \begin{bmatrix} \textcircled{11} / \textcircled{15} & \textcircled{12} / \textcircled{16} \\ \textcircled{13} / \textcircled{17} & \textcircled{14} / \textcircled{18} \end{bmatrix}$	
Z _{REF} ⇒ $\textcircled{2}$	INDEX ⇒ $\textcircled{34}$
Frequency ⇒ $\textcircled{20}$ Step Freq ⇒ $\textcircled{21}$ Step Freq ⇒ $\textcircled{22}$	



















Flags

Flag 0 ⇒ Ladder Analysis	Flag 1 ⇒ Shunt Element
Flag 24 ⇒ Set by program	
Flag 4 ⇒ Auto execution of CHR & STB	

ELEMENT LIBRARY (S= S_p)

LEVEL	FLAG 1 = 0	FLAG 2 = 1	DATA
PLC			L, C
PRZ			R, L, C
TRL			R, I
OST			R, I
SST			R, I
TRF			R ₁ , R ₂ R, R
CES			R, ⊕ R, ⊕
VCS			P, ⊕ or Im, ⊕
CPL		SET FLAG 2, 3, 4 FOR OUTPUT INFL R ₁ ⇒ ⊕ ⊕ R ₂ ⇒ ⊕ ⊕ R & E	Zoc, Ie, Zoo, Ie

ELEMENT LIBRARY (S= S_B)

LABEL	FLAG 1 = 0	FLAG 1 = 1	DATA
RES			R
CAP			C
IND			L
SRL			R, L
SRC			R, C
SLE			L, C
SRZ			R, L, C
PRL			R, L
PRC			R, C

INTERCONNECTIONS

CAS		
SER		
PAR		
SFC		
PSC		
HYB		
PBR	<p style="text-align: center;"><u>ENTER Z_L, Q_L</u></p>	<p style="text-align: center;"><u>ENTER Z_L, Q_L</u></p>
		<p style="text-align: center;">SET FLAG 2</p>
SBR	<p style="text-align: center;"><u>ENTER Z_L, Q_L</u></p>	<p style="text-align: center;"><u>ENTER Z_L, Q_L</u></p>
		<p style="text-align: center;">SET FLAG 2</p>

LADDER ANALYSIS

SF 00 FOR LADDER ANALYSIS MODE.

FREQ \rightarrow (20); STEP FREQ. \rightarrow (21); STEP FAER. \rightarrow (22);

|Z| \rightarrow (23); $\theta_Z \rightarrow$ (24).

S_A IS NOT MODIFIED

• ALL R-L-C, TRN, DST, & SST ELEMENTS AVAILABLE.

DATA INPUT & FLAGS IDENTICAL TO Z-FORT ANALYSIS.

• STORE ZL IN (25) & (26). ENTER AND EXECUTE ELEMENT.

Z_{in} IN (25) & (26) AS WELL AS "X" AND "Y".

MANIPULATION COMMANDS

SVA	$S_n \Rightarrow S_m$	<u>ENTER M</u> SIZE = 25 + 8·N
SVB	$S_n \Rightarrow S_m$	<u>ENTER M</u> SIZE = 25 + 8·N
USA	$S_n \Leftarrow S_m$	<u>ENTER M</u> SIZE = 25 + 8·N
USB	$S_n \Leftarrow S_m$	<u>ENTER M</u> SIZE = 25 + 8·N

EAC	$S_n \leftrightarrow S_m$	
EQV	$S_n \Rightarrow S_n$	
CHA		

CONVERSION COMMANDS

X12		
S-Y	$S_A \Rightarrow Y_A$	
Z-S	$Z_A \Rightarrow S_A$	
S-Y	$S_A \Rightarrow Y_A$	
Y-S	$Y_A \Rightarrow S_A$	
S-H	$S_A \Rightarrow H_A$	
H-S	$H_A \Rightarrow S_A$	
Z-R	$Z \Rightarrow \Gamma$	ENTER Z, ⊖
R-Z	$\Gamma \Rightarrow Z$	ENTER P, ⊖

X23		
S-G	$S_A \Rightarrow G_A$	
G-S	$G_A \Rightarrow S_A$	
Y-Z	$Y_A \Rightarrow Z_A$	
Z-Y	$Z_A \Rightarrow Y_A$	
H-G	$H_A \Rightarrow G_A$	
G-H	$G_A \Rightarrow H_A$	
Y-R	$Y_{mnhk} \Rightarrow \Gamma$	ENTER Y, ⊖
R-Y	$\Gamma \Rightarrow Y_{mnhk}$	ENTER P, ⊖

CHARACTERIZATION: S_A

CHR	<p><u>K<1</u> output : K, WCC</p> <p><u>K<2</u> output : K, G_{max}</p> <p><u>END</u> : Fin, G_{max}, F_{max}, G_{max}</p> <p><u>STOP</u> : Return without stop</p> <p>K = ① ; G_{max} (max) = ② ; F_{max} = ③</p> <p>G_{max} = ④ ; F_{max} = ⑤ ; G_{max} = ⑥</p>
GTF	<p><u>ENTER</u> : I₀₁, W₀₁, I₀₁, W₀₁</p> <p><u>ENTER</u> : G₀ (dB)</p>
CBC	<p>at "2", <u>ENTER</u> { <u>R/S</u> to end : for input value : for output value</p> <p>at "G", <u>ENTER</u> , G_{max} (dB)</p> <p><u>OUTPUT</u> : G_{max}, I_{ctrl}, G_{max}, W_{max}</p> <p><u>R/S</u></p> <p>at "G", <u>ENTER</u> { <u>ENTER</u> to end <u>R/S</u> go to "2"</p>
LBL 29	<p><u>ENTER</u> : I₀₁, W₀₁</p> <p>at "L", <u>ENTER</u> { <u>R/S</u> to end : Loss (dB)</p> <p><u>OUTPUT</u> : Loss, I_{ctrl}, G_{max}, RAD</p> <p><u>R/S</u> : go to "L"</p>
LBL 30	<p><u>ENTER</u> : I₀₁, W₀₁</p> <p>at "L", <u>ENTER</u> { <u>R/S</u> to end : Loss (dB)</p> <p><u>OUTPUT</u> : Loss, I_{ctrl}, G_{max}, RAD</p> <p><u>R/S</u> : go to "L"</p>

CHARACTERIZATION: S_A

<p>STB</p>	<p>AT "?", <u>ENTER</u> $\left\{ \begin{array}{l} \boxed{R/S} \text{ to end} \\ 1 \text{ for INPUT PLANE} \\ 2 \text{ for OUTPUT PLANE} \end{array} \right.$</p> <p><u>OUTPUT</u>: $CTA , \theta_{CTA}, RAD, \text{Code}$</p> <p>$\boxed{R/S}$: 90 & " ?"</p> <hr/> <p><u>SF 04</u>: Return without Stop</p> <p><u>ENTER</u> $\left\{ \begin{array}{l} 1 \text{ for INPUT PLANE} \\ 2 \text{ for OUTPUT PLANE} \end{array} \right.$</p> <p>$K \leftarrow (29)$; $CTR1 \leftarrow (31)$; $\theta_{CTR1} \leftarrow (32)$;</p> <p>$RAD \leftarrow (33)$; $CODE \leftarrow (30)$.</p> <hr/> <p><u>CODE</u>: $\left\{ \begin{array}{l} "0" \text{ STABLE OUTSIDE} \\ "1" \text{ STABLE INSIDE} \\ "10" \text{ UNCONDITIONAL STABILITY} \end{array} \right.$</p>
<p>GTP</p>	<p><u>ENTER</u>: $R_1 , \theta_T, R_2 , \theta_L$</p> <p><u>OUTPUT</u>: G_T (dB)</p>
<p>S1P</p>	<p><u>ENTER</u>: R_1 , θ_L</p> <p><u>OUTPUT</u>: S_{11}' , θ_{11}'</p>
<p>S2P</p>	<p><u>ENTER</u>: R_1 , θ_S</p> <p><u>OUTPUT</u>: S_{22}' , θ_{22}'</p>
<p>MAL</p>	<p><u>ENTER</u>: $R_1 , \theta_1, R_2 , \theta_2$</p> <p><u>OUTPUT</u>: MISMATCH LOSS (dB)</p>

MAPPING: SA

<p>MAP</p> <p>LSL N</p>	<p>AT "F", ENTER</p> <ul style="list-style-type: none"> 1 for SMITH CHART 2 for MAPSOURCE 3 for MAPLOAD 4 for FORM FEEDBACK 5 for FORM FEEDBACK <p>FORMS OR FORM FEEDBACK:</p> <p>AT "G" $\frac{1}{2}$, ENTER $\frac{1}{2}$ AND MAP OF $\frac{1}{2}$</p>
-------------------------	--

<p>AT "MAP?", ENTER</p> <p>AS "R" (or "X"), ENTER value in 3 units</p> <p>MAP: R(-X), FORM, OVER, RAD</p> <p>$\frac{1}{2}$: y to "MAP?"</p>	<ul style="list-style-type: none"> R for Contact RESEARCH X for CONTACT RESEARCH S for SMITH CHART <p>$\frac{1}{2}$ to end</p>
---	--

© COPYRIGHT 1982, COMSAT General Integrated Systems, Inc. All rights reserved.
No part of this document may be copied or reproduced in any manner without the
express written permission of COMSAT General Integrated Systems Inc.



Comsat General Integrated Systems, Inc.
Compact Division

1131 San Antonio Road • Palo Alto, California 94303 • (415) 966-8440
7801 North Lamar • Austin, Texas 78752 • (512) 451-7938